# Maintainer: Lukas Fleischer <lfleischer@archlinux.org>

pkgname=unhide
pkgver=20220611
pkgrel=2
pkgdesc='A forensic tool to find processes hidden by rootkits, LKMs or by other techniques.'
arch=('loong64' 'x86_64')
url='https://github.com/YJesus/Unhide/'
license=('GPL3')
depends=('iproute2' 'lsof' 'psmisc' 'net-tools' 'procps-ng')
source=("https://github.com/YJesus/Unhide/archive/v$pkgver/$pkgname-$pkgver.tar.gz"
        "Makefile")
md5sums=('cb0fc465ef26e907ba2166551dc27369'
         '68ec8dcd2f25d4d5dad089a8535d462f')
sha256sums=('db0c29b6519982df07a9024d37b1e4618d40eaf13c2c01737ec79dcfba108eb6'
            '6d2a37133a5cb05995bf0c8f8230820ce24b6f3369c49e84a6da7684d2bb6579')

prepare() {
  cp -f Makefile Unhide-$pkgver/
}

build() {
  cd Unhide-$pkgver
  make
}

package() {
  cd Unhide-$pkgver
  make PREFIX=/usr DESTDIR="$pkgdir" install
  ln -s unhide-linux "$pkgdir/usr/bin/unhide"
}
